Introduction to Netlogo

Agent-based simulation

Netlogo

Modeling complex systems

- Programmable modeling environment for simulating natural and social phenomena
  - Well suited for modeling complex systems evolving over time
  - Hundreds or thousands of independent agents operating concurrently
  - Exploring the connection between the micro-level behavior of individuals and the macro-level patterns that emerge from the interaction of many individuals

Netlogo
Modeling complex systems

- Easy-to-use application development environment
  - creating custom models and quickly testing hypotheses about self-organized systems
  - simple scripting language
  - user-friendly graphical interface
  - runs on JVM

Wilensky, U. (1999). NetLogo. http://ccl.northwestern.edu/netlogo/. Center for Connected Learning and Computer-Based Modeling, Northwestern University, Evanston, IL

Netlogo

Practical Info

- Download link:
  - http://ccl.northwestern.edu/netlogo/download.shtml
  - Launch Netlogo through command line:
    - $ /{netlogo_download_folder}/netlogo.sh
- Online doc:
  - https://ccl.northwestern.edu/netlogo/docs/
- Book for agent-based modeling (special focus on Netlogo):
  - http://www.intro-to-abm.com/

Netlogo
History snapshot

- LOGO (Papert & Minsky, 1967)
  - theory of education based on Piaget's constructionism ("hands-on" creation and test of concepts)
  - simple language derived from LISP
  - turtle graphics and exploration of "micro-worlds"
- StarLogo (Resnick, 1991), MacStarLogo, StarLogoT
  - agent-based simulation language
- NetLogo (Wilensky, 1999)
  - further extending StarLogo (continuous turtle coordinates, cross-platform, networking, etc.)

Netlogo

The world of Netlogo

- NetLogo is a 2-D world made of 4 kinds of agents:
  - *Patches* - make up the background or “landscape”
  - *Turtles* - move around on top of the patches
  - *Links* - connect two turtles
  - *The Observer* - oversees everything going on in the world

Graphical Interface

Controls

- Controls allow to run and manage the flow of execution
  - Buttons: initialize, start, stop, step through the model

- "Once" button execute one action
- "Forever" button repeat the same action until pressed again

  - Functions with the name of the buttons specify the action executed on click
  - Command centre: ask agents to execute specific commands "on the fly"

Graphical Interface

Settings

- Settings allow to modify parameters
  - Sliders: adjust a quantity from *min* to *max* by an *increment*

population = 126

  - Switches: set a Boolean variable

incentivi_installazione? = false

  - Choosers: set a value from a list



file = “Beats/seth2.csv”

Graphical Interface

Views

- Views allow to display information
  - Monitors display the current value of variables



  - Plots: display the history of a variable's value



  - Output text areas, log text info

Graphical Interface

Views

- Graphic window, the main view of the 2-D Netlogo world

Model evolution (based on discrete time-steps)



Programming Concepts
Agents

- Agents carry out their activity, all simultaneously
  - Patches don't move, form a 2-D wrap-around grid, have integer coordinates (*pxcor*,*pycor*)
  - Turtles move on top of patches (not necessarily in their centre), have decimal coordinates (*xcor*,*ycor*) and orientation (*heading*)
  - Observer can create new turtles, can have read/write access to all the agents and variables

Programming Concepts

Procedures & Functions

- Commands ("to" keyword)
  - Action for the agents to carry out ("void" functions)
  - Example with 2 input arguments:

```
to draw-polygon [ num-sides size ]
      pd     ;; pen down, draw
      repeat num-sides
      [      fd size    ;; forward 'size' steps
             rt (360 / num-sides) ]   ;; rotate
end
```

Programming Concepts

Procedures & Functions

- Reporters ("to-report")
  - Report a result value
  - Example with 1 input arguments:

```
to-report absolute-value [ number ]
      if-else number >= 0
      [ report number ]
      [ report 0 - number ]
end
```

- Primitives
  - *Built-in* command or reporters
  - Some have an abbreviated form (create-turtle <--> crt)
- Procedures
  - *Custom* commands or reporters (user made)

Programming Concepts
Variables

- Variables – places to store values
  - Global variables: only one value for the variable and every agent can access it
  - Turtle and Patch variables: each turtle/patch has its own value for every turtle/patch variable
  - Local variables: defined and accessible only inside a procedure (scope=narrowest square brackets or procedure itself)

Programming Concepts

Variables

- Built-in variables
  - Ex. turtle variables: color, xcor, ycor, etc.
  - Ex. Patch variables: pcolor, pxcor, etc.
- Custom variables
  - Defining global variables

```
global [ clock ]
```

  - Defining turtle/patch variables

```
turtles-own [ energy speed ]
patches-own [ friction ]
```

Programming Concepts
Variables

- Custom variables
  - Defining global variables
  - Defining turtle/patch variables
  - Defining local variables:
    - **`let variable value`**
    - Creates a new local variable and gives it the desired value

```
to swap-colors [ t1 t2 ]
      let temp color-of t1
```

  - Setting a variable values (after its definition):
    - **`set variable value`**

Programming Concepts
Ask

- *Ask* – specify commands to be run by turtles or patches
  - Asking all turtles

```
ask turtles [ … ]
```

  - Asking all patches
  - Asking *N* turtles

```
ask n-of N turtles [ … ]
```

- Observer code *cannot* be inside any "ask" block

Programming Concepts
Variables

- Setting variables
  - Setting the color of all turtles

```
ask turtles[ set color red ]
```

  - Setting the color of all patches

```
ask patches[ set pcolor red ]
```

  - Setting the color of the patches under the turtles

```
ask turtles [ set pcolor red ]
```

  - Setting the color of one turtle (identify by ID)

```
ask turtle 5 [ set color green]
```

  - or

```
set color-of turtle 5 green
```

  - Setting the color of one patch (identified with coordinates)

```
ask patch 2 3 [ set pcolor green ]
```

Programming Concepts

Agent sets

- Agent set, definition of a subset of agents (not a keyword)
  - All blue turtles

```
turtles with [ color = blue ]
```

  - All blue turtles on the patch of the current caller (patch or turtle)

```
turtles-here with [ color = blue ]
```

  - All turtles less than 5 patches away from caller

```
turtles in-radius 5
```

  - The 4 patches to the east, north, west and south of the caller

```
patches at-points [[1 0] [0 1] [-1 0] [0 -1]]
```

Programming Concepts

Agent sets

- Using agent sets
  - Ask such agents to execute a command

    ```
    ask <agentset> [ … ]
    ```

  - Check if there are such agents

    ```
    show any? <agentset>
    ```

  - Count such agents

    ```
    show count <agentset>
    ```

- Ex. - remove the richest turtle (with the maximum “assets” value):

  ```
  ask max-one-of turtles [sum assets] [die]
  ```

  agentset — command/action

Programming Concepts

Breeds

- *Breed*, a “natural” kind of agent set (other species than turtle)

```
breed [ wolves sheep ]
```

- A new breed comes with automatically derived primitives:

```
create-<breed>
create-custom-<breed>
<breed>-here
<breed>-at
```

- The breed is a turtle variable

```
ask turtles 5 [ if breed=sheep ]
```

- A turtle agent can change breed

```
ask turtles 5 [ set breed sheep ]
```

Exercise 1

Basic Ants Model

- Very simple model as a first "hands-on" experience
- A colony of ants forages for food
  - Tough each ant follows a set of simpler rules, the colony as a whole act in a sophisticated way



Wilensky, U. (1997). NetLogo Ants model. http://ccl.northwestern.edu/netlogo/models/Ants. Center for Connected Learning and Computer-Based Modeling, Northwestern University, Evanston, IL.

How to start

```
netlogo-5.0.2 — java ◂ netlogo.sh — 113×30
Last login: Thu Oct 29 20:03:37 on ttys005
(base) MBP-di-Allegra:~ allegradefilippo$ cd /Users/allegradefilippo/Desktop/esercitazioneNetlogo/netlogo-5.0.2
(base) MBP-di-Allegra:netlogo-5.0.2 allegradefilippo$ ls
HubNet.jar              docs                    lib                     netlogo.sh
Mathematica Link        extensions              models                  netlogo.sh~
NetLogo.jar             hs_err_pid11496.log     netlogo-3D.sh           netlogo_logging.xml
NetLogoLite.jar         hubnet.sh               netlogo-headless.sh     readme.txt
NetLogoLite.jar.pack.gz icon.ico                netlogo-headless.sh~
(base) MBP-di-Allegra:netlogo-5.0.2 allegradefilippo$ ./netlogo.sh
Java HotSpot(TM) 64-Bit Server VM warning: ignoring option MaxPermSize=128m; support was removed in 8.0
```



How to start


NetLogo
Interface
Info
Code
Edit
Delete
Add
Button
normal speed
view updates
continuous
Settings...
ticks:
3D
Command Center
Clear
observer>
netlogo.sh — 113×30
adefilippo/Desktop/esercitazioneNetlogo/netlogo-5.0.2
netlogo.sh
netlogo.sh~
.sh
netlogo_logging.xml
adless.sh
readme.txt
adless.sh~
tlogo.sh
MaxPermSize=128m; support was removed in 8.0

How to start



1. Open an existing model .nlogo



2. Choose the model of the first exercise (i.e.) Es_1_AntsModel.nlogo

Es_1_AntsModel.nlogo



Exercise 1

Ants Model

- When an ant finds a piece of food, it carries the food back to the nest, dropping a chemical as it moves
- When other ants “sniff” the chemical, they follow the chemical toward the food
- As more ants carry food to the nest, they reinforce the chemical trail

Es_1_AntsModel.nlogo



Exercise 1

Model Usage

- Click the SETUP button to set up the ant nest (in violet, at center) and three piles of food then click the GO button to start the simulation.
  - The chemical is shown in a green-to-white gradient.

- The EVAPORATION-RATE slider controls the evaporation rate of the chemical. The DIFFUSION-RATE slider controls the diffusion rate of the chemical.
- If you want to change the number of ants, move the POPULATION slider before pressing SETUP

Es_1_AntsModel.nlogo

NetLogo — Es1_AntsModel {/Users/allegradefilippo/Desktop/esercitazioneNetlogo}

Interface | Info | Code

Find... | Check | Procedures | Indent automatically

```
patches-own [
  chemical             ;; amount of chemical on this patch
  food                 ;; amount of food on this patch (0, 1, or 2)
  nest?                ;; true on nest patches, false elsewhere
  nest-scent           ;; number that is higher closer to the nest
  food-source-number   ;; number (1, 2, or 3) to identify the food sources
]

;;;;;;;;;;;;;;;;;;;;;;;;
;;; Setup procedures ;;;
;;;;;;;;;;;;;;;;;;;;;;;;

to setup
  clear-all
  set-default-shape turtles "bug"
  crt population
  [ set size 2         ;; easier to see
    set color red  ]   ;; red = not carrying food
  setup-patches
  reset-ticks
end

to setup-patches
  ask patches
  [ setup-nest
    setup-food
    recolor-patch ]
end

to setup-nest  ;; patch procedure
  ;; set nest? variable to true inside the nest, false elsewhere
  set nest? (distancexy 0 0) < 5
  ;; spread a nest-scent over the whole world -- stronger near the nest
  set nest-scent 200 - distancexy 0 0
end
```

Es_1_AntsModel.nlogo

NetLogo — Es1_AntsModel {/Users/allegradefilippo/Desktop/esercitazioneNetlogo}

Interface | Info | Code

Find... | Check | Procedures | Indent automatically

```
to setup-food  ;; patch procedure
  ;; setup food source one on the right
  if (distancexy (0.6 * max-pxcor) 0) < 5
  [ set food-source-number 1 ]
  ;; setup food source two on the lower-left
  if (distancexy (-0.6 * max-pxcor) (-0.6 * max-pycor)) < 5
  [ set food-source-number 2 ]
  ;; setup food source three on the upper-left
  if (distancexy (-0.8 * max-pxcor) (0.8 * max-pycor)) < 5
  [ set food-source-number 3 ]
  ;; set "food" at sources to either 1 or 2, randomly
  if food-source-number > 0
  [ set food one-of [1 2] ]
end

to recolor-patch  ;; patch procedure
  ;; give color to nest and food sources
  ifelse nest?
  [ set pcolor violet ]
  [ ifelse food > 0
    [ if food-source-number = 1 [ set pcolor cyan ]
      if food-source-number = 2 [ set pcolor sky  ]
      if food-source-number = 3 [ set pcolor blue ] ]
    ;; scale color to show chemical concentration
    [ set pcolor scale-color green chemical 0.1 5 ] ]
end
```

Es_1_AntsModel.nlogo




```
;;;;;;;;;;;;;;;;;;;;;;
;;; Go procedures ;;;
;;;;;;;;;;;;;;;;;;;;;;

to go  ;; forever button
  ask turtles
  [ if who >= ticks [ stop ] ;; delay initial departure
    ifelse color = red
    [ look-for-food  ]       ;; not carrying food? look for it
    [ return-to-nest ]       ;; carrying food? take it back to nest
    wiggle
    fd 1 ]
  diffuse chemical (diffusion-rate / 100)
  ask patches
  [ set chemical chemical * (100 - evaporation-rate) / 100  ;; slowly evaporate chemical
    recolor-patch ]
  tick
end

to return-to-nest  ;; turtle procedure
  ifelse nest?
  [ ;; drop food and head out again
    set color red
    rt 180 ]
  [ set chemical chemical + 60  ;; drop some chemical
    uphill-nest-scent ]         ;; head toward the greatest value of nest-scent
end

to look-for-food  ;; turtle procedure
  if food > 0
  [ set color orange + 1     ;; pick up food
    set food food - 1        ;; and reduce the food source
    rt 180                   ;; and turn around
    stop ]
  ;; go in the direction where the chemical smell is strongest
  if (chemical >= 0.05) and (chemical < 2)
  [ uphill-chemical ]
end
```

Es_1_AntsModel.nlogo

NetLogo — Es1_AntsModel {/Users/allegradefilippo/Desktop/esercitazioneNetlogo}

Interface | Info | Code

Find... | Check | Procedures | Indent automatically

```
  [ uphill-chemical ]
end

;; sniff left and right, and go where the strongest smell is
to uphill-chemical  ;; turtle procedure
  let scent-ahead chemical-scent-at-angle   0
  let scent-right chemical-scent-at-angle  45
  let scent-left  chemical-scent-at-angle -45
  if (scent-right > scent-ahead) or (scent-left > scent-ahead)
  [ ifelse scent-right > scent-left
    [ rt 45 ]
    [ lt 45 ] ]
end

;; sniff left and right, and go where the strongest smell is
to uphill-nest-scent  ;; turtle procedure
  let scent-ahead nest-scent-at-angle   0
  let scent-right nest-scent-at-angle  45
  let scent-left  nest-scent-at-angle -45
  if (scent-right > scent-ahead) or (scent-left > scent-ahead)
  [ ifelse scent-right > scent-left
    [ rt 45 ]
    [ lt 45 ] ]
end

to wiggle  ;; turtle procedure
  rt random 40
  lt random 40
  if not can-move? 1 [ rt 180 ]
end

to-report nest-scent-at-angle [angle]
  let p patch-right-and-ahead angle 1
  if p = nobody [ report 0 ]
  report [nest-scent] of p
end

to-report chemical-scent-at-angle [angle]
  let p patch-right-and-ahead angle 1
  if p = nobody [ report 0 ]
  report [chemical] of p
end
```

Exercise 1

Things to notice

- The ant colony generally exploits the food source in order, starting with the food closest to the nest, and finishing with the food most distant from the nest
- It is more difficult for the ants to form a stable trail to the more distant food, since the chemical trail has more time to evaporate and diffuse before being reinforced

Exercise 1

Things to notice

- Once the colony finishes collecting the closest food, the chemical trail to that food naturally disappears, freeing up ants to help collect the other food sources.
  - The more distant food sources require a larger “critical number” of ants to form a stable trail.
- The consumption of the food is shown in a plot.
  - The line colors in the plot match the colors of the food piles.

Exercise 1

Model Extensions

1. Try different placements for the food sources
   - What happens if two food sources are equidistant from the nest?
2. In this project, the ants use a “trick” to find their way back to the nest: they follow the “nest scent.”
   - Real ants use a variety of different approaches to find their way back to the nest.
   - Try to implement some alternative strategies.
3. In the uphill-chemical procedure, the ant “follows the gradient” of the chemical. That is, it “sniffs” in three directions, then turns in the direction where the chemical is strongest.
   - Try variants of the uphill-chemical procedure, changing the number and placement of “ant sniffs.”

Exercise 2

Ant Colony Optimization e TSP

- Goal: implementing the Ant System algorithm[1] and use it to solve the Traveling Salesman Problem
- Based on the observation of ants behaviour
  - Positive feedback based on pheromone tracks which reinforce the best solution components



[1] *Dorigo, M., Maniezzo, V., and Colorni, A., The Ant System: Optimization by a colony of cooperating agents. IEEE Transactions on Systems, Man, and Cybernetics Part B: Cybernetics, Vol. 26, No. 1. (1996), pp. 29-41. http://citeseer.ist.psu.edu/dorigo96ant.html*

Exercise 2

Ant Colony

- As you already know
  - Ants leave a pheromone trail while going from the nest to food sources (and vice versa)
  - Ants tend to choose (with higher probability) routes with greater amount of pheromones
  - Cooperative interaction which leads to an emergent behaviour, that is finding the shortest path

Exercise 2

ACO

- Probabilistic model (*pheromone* model) used to recreate the pheromone trails left by ants
- Ants incrementally build the component of a solution
- Ants perform stochastic steps on a fully connected graph (*construction graph*)
- Constraints used to obtain a feasible solution

Exercise 2

ACO and TSP

- A possible model for TSP:
  - The graph nodes are the city to visit (the components of a solution)
  - The edges are the connections between the cities
  - A solution is a Hamiltonian circuit in the graph
  - Constraints are used to avoid loops, so that an ant can visit a city exactly once

Exercise 2

Information sources

- Edges or vertexes (or both) have two information:
  - Pheromone τ, which stands in for natural trail left by ants and represents the long term memory of ants in relation to the global search process
  - Heuristic value η, i.e. the a priori knowledge on the problem

Exercise 2

ACO System

- The ants follow a path on the construction graph and build a solution
- They used a transition (probabilistic) rule to choose the next node to visit
- Both pheromone and heuristic are taken into account
- Pheromone values are adjusted based on the quality of the solution found